

(Local) In[135]:=

```
AppendTo[$Path, NotebookDirectory[]];  
<< Matte`
```

Matte package loaded. © 2017–2018 Jonas Einarsson (me@jonaseinarsson.se)

Released under the MIT License <https://opensource.org/licenses/MIT>

Important!!!

Matte unitsphereIntegral only works for 3D! I made a separate one for 2D, make sure you've loaded the right one.

Tensor Spherical Harmonics are made up of *irreducible* rank-n tensors

(Local) In[137]:=

```
On[Assert];  
Assert[trace[delta] == 3]
```

These are described nicely in Leal Advanced Transport Phenomena Pg. 527 (within Chapter 8 Creeping Flows)

They can be developed from the solutions to the Laplace equation in spherical coordinates. Where you are taking repeated gradients of $(1/r)$.

I.E. $\nabla(\nabla(1/r))$ not Laplacian but a gradient of a gradient.

That should be correct, but constants are really important and magnitude of r is equal to 1, and so it is often written in a different method.

$$\phi_{-(n+1)} = \frac{(-1)^n}{(2n-1)!!} \nabla_1 \dots \nabla_n \left(\frac{1}{r} \right), \quad n = 0, 1, 2, \dots$$

That technically gets you the $-(n+1)$ th decaying harmonic, but then the n th order irreducible tensor just has you substitute $|r| = 1$

First use a recurrence relation to get the repeated gradient of $\frac{1}{r}$ where $\text{invr}[1]$ is $\frac{1}{r}$

(Local) In[139]:=

```
RepeatedPartialR[0] := invr[1];
RepeatedPartialR[n_] := partialr[RepeatedPartialR[n - 1], idx[n]] // contract;
```

Now use that to get the decaying harmonics and replace inverse r to any power with 1

(Local) In[141]:=

```
IrreducibleRanknTensor[n_] :=
  ReplaceAll[invr[x_] → 1] [  $\frac{(-1)^n}{(2 * n - 1)!!}$  RepeatedPartialR[n] // contract ];
```

So for *any* vector r, the irreducible rank-n tensor that describes its dyad

(Local) In[142]:=

```
IrreducibleRanknTensor[0] // prettyprint
IrreducibleRanknTensor[1] // prettyprint
IrreducibleRanknTensor[2] // prettyprint
IrreducibleRanknTensor[3] // prettyprint
IrreducibleRanknTensor[4] // prettyprint
```

(Local) Out[142]=

1

(Local) Out[143]=

 r_i

(Local) Out[144]=

$$r_i r_j - \frac{\delta_{i,j}}{3}$$

(Local) Out[145]=

$$r_i r_j r_k - \frac{1}{5} r_k \delta_{i,j} - \frac{1}{5} r_j \delta_{i,k} - \frac{1}{5} r_i \delta_{j,k}$$

(Local) Out[146]=

$$r_i r_j r_k r_l - \frac{1}{7} r_k r_l \delta_{i,j} - \frac{1}{7} r_j r_l \delta_{i,k} - \frac{1}{7} r_j r_k \delta_{i,l} - \frac{1}{7} r_i r_l \delta_{j,k} +$$

$$\frac{1}{35} \delta_{i,l} \delta_{j,k} - \frac{1}{7} r_i r_k \delta_{j,l} + \frac{1}{35} \delta_{i,k} \delta_{j,l} - \frac{1}{7} r_i r_j \delta_{k,l} + \frac{1}{35} \delta_{i,j} \delta_{k,l}$$

Here are the repeated gradients of 1/r to show you that they are equivalent, just divided by leading constants.

(Local) In[147]:=

```

x1 = partialr[invr[1], idx[1]] // contract;
x2 = partialr[x1, idx[2]] // contract;
x3 = partialr[x2, idx[3]] // contract;
x1 // prettyprint
x2 // prettyprint
x3 // prettyprint

```

(Local) Out[150]=

$$-\frac{r_i}{r^3}$$

(Local) Out[151]=

$$\frac{3 r_i r_j}{r^5} - \frac{\delta_{i,j}}{r^3}$$

(Local) Out[152]=

$$-\frac{15 r_i r_j r_k}{r^7} + \frac{3 r_k \delta_{i,j}}{r^5} + \frac{3 r_j \delta_{i,k}}{r^5} + \frac{3 r_i \delta_{j,k}}{r^5}$$

Part 2: Using as basis functions.

Assert

$$f(x, q, t) = [c_0 n(x, t) * T_0 + c_1 m_\alpha(x, t) * T_{1\alpha} + c_2 Q_{\alpha,\beta}(x, t) * T_{2\alpha,\beta} + c_3 C_{\alpha,\beta,\gamma}(x, t) T_{3\alpha,\beta,\gamma} + \dots]$$

Where T_n is the rank n irreducible tensor harmonic.

Note that there are competing definitions for \mathbf{Q} ., one is the traceless version, the other is not. Have to be careful and look at the definition. For this document will be using the traceless version. $\mathbf{Q} =$

$$\int (\mathbf{q}\mathbf{q} - \frac{1}{3}) f d\mathbf{q}$$

Also be careful, some people include the prefactor $\frac{1}{4\pi}$ in the integral, others do not. For this document, we will not.

Now you find the constants by recognizing that the tensor harmonics form an orthogonal basis set.

Get c0

$$\int f(x, q, t) T_0 d\mathbf{q} = \int T_0 * [c_0 n * T_0 + c_1 m_\alpha * T_{1\alpha} + c_2 Q_{\alpha,\beta} * T_{2\alpha,\beta} + c_3 C_{\alpha,\beta,\gamma} T_{3\alpha,\beta,\gamma} + \dots] d\mathbf{q}$$

$$\int f(x, q, t) T_0 \, dq = c_0 n \int (T_0 * T_0) \, dq$$

On our left hand side is our definition for n

$$n = \int (c_0 n * T_0 * T_0) \, dq$$

Knowing $T_0 = 1$. We perform the unit sphere integral and see:

(there is a way to do the unit sphere integrals by hand, see attached pdf or this link , but I'll let matte handle it)

(Local) In[153]:=

```
unitsphereIntegral[c0 nfield * IrreducibleRanknTensor[0] * IrreducibleRanknTensor[0]]
```

(Local) Out[153]=

```
4 c0 nfield π
```

$$n = \int (c_0 n * T_0 * T_0) \, dq = 4 \pi c_0 n$$

$$c_0 = \frac{1}{4 \pi}$$

Get c1

$$\int f(x, q, t) T_{1n} \, dq = \int T_{1n} * [c_0 n * T_0 + c_1 m_\alpha * T_{1\alpha} + c_2 Q_{\alpha,\beta} * T_{2\alpha,\beta} + c_3 C_{\alpha,\beta,\gamma} T_{3\alpha,\beta,\gamma} + \dots] \, dq$$

$$\int f(x, q, t) T_{1n} \, dq = c_1 m_\alpha \int (T_{1\alpha} * T_{1n}) \, dq$$

On our left hand side is our definition for m

$$m_n = c_1 m_\alpha \int (T_{1\alpha} * T_{1n}) \, dq$$

For convenience we are going to rename indices $\alpha = i, n = j$

$$m_j = c_1 m_i \int (T_{1i} * T_{1j}) \, dq$$

Knowing $T_1 = q$. We perform the unit sphere integral and see:

(Local) In[154]:=

```
(*The indices need to be different! Set it up that way*)
T1alpha = IrreducibleRanknTensor[1]
T1n = IrreducibleRanknTensor[1] /. idx[1] -> idx[2]
unitsphereIntegral[ c1 mfield T1alpha * T1n // Expand // contract] // prettyprint
```

(Local) Out[154]=

$$r[idx[1]]$$

(Local) Out[155]=

$$r[idx[2]]$$

(Local) Out[156]=

$$\frac{4}{3} c1 mfield \pi \delta_{i,j}$$

$$m_j = c_1 m_i * \frac{4}{3} \pi \delta_{i,j}$$

Contract the indices and solve

$$m_j = c_1 m_j * \frac{4}{3} \pi$$

$$c_1 = \frac{3}{4\pi}$$

Last example. Find c2. The **trickiest**!

$$\int f(x, q, t) T_{2n,m} dq = \int T_{1n} * [c_0 n * T_0 + c_1 m_\alpha * T_{1\alpha} + c_2 Q_{\alpha,\beta} * T_{2\alpha,\beta} + c_3 C_{\alpha,\beta,\gamma} T_{3\alpha,\beta,\gamma} + \dots] dq$$

$$\int f(x, q, t) T_{2n,m} dq = c_1 Q_{\alpha,\beta} * \int (T_{2\alpha,\beta} * T_{1n,m}) dq$$

On our left hand side is our definition for m

For convenience we are going to rename indices $\alpha = i, \beta = j, n = k, m = l$

$$Q_{k,l} = c_2 Q_{i,j} * \int (T_{2k,l} * T_{2i,j}) dq$$

Knowing $T_2 = qq - \delta/3$. We perform the unit sphere integral and see:

(Local) In[157]:=

```
(*The indices need to be different! Set it up that way*)
T2ij = IrreducibleRanknTensor[2]
T2kl = IrreducibleRanknTensor[2] /. idx[1] -> idx[3] /. idx[2] -> idx[4]
unitsphereIntegral[c2 Qfield T2ij * T2kl // Expand // contract] // prettyprint
unitsphereIntegral[c2 Qfield[idx[1], idx[2]] T2ij * T2kl // Expand // contract] // contract //
prettyprint
```

(Local) Out[157]=

$$-\frac{1}{3} \text{delta}[\text{idx}[1], \text{idx}[2]] + r[\text{idx}[1]] \times r[\text{idx}[2]]$$

(Local) Out[158]=

$$-\frac{1}{3} \text{delta}[\text{idx}[3], \text{idx}[4]] + r[\text{idx}[3]] \times r[\text{idx}[4]]$$

(Local) Out[159]=

$$\frac{4}{15} c_2 \pi Q_{\text{field}} \delta_{i,l} \delta_{j,k} + \frac{4}{15} c_2 \pi Q_{\text{field}} \delta_{i,k} \delta_{j,l} - \frac{8}{45} c_2 \pi Q_{\text{field}} \delta_{i,j} \delta_{k,l}$$

(Local) Out[160]=

$$\frac{4}{15} c_2 \pi Q_{\text{field}} \delta_{k,l} + \frac{4}{15} c_2 \pi Q_{\text{field}} \delta_{l,k} - \frac{8}{45} c_2 \pi \delta_{k,l} \text{trace}[Q_{\text{field}}]$$

$$Q_{k,l} = c_2 Q_{ij} * \left[\frac{4}{15} \pi \delta_{i,l} \delta_{j,k} + \frac{4}{15} \pi \delta_{i,k} \delta_{j,l} - \frac{8}{45} \pi \delta_{ij} \delta_{k,l} \right]$$

$$Q_{k,l} = c_2 \frac{4\pi}{15} * \left[Q_{k,l} + Q_{l,k} - \frac{2}{3} \delta_{k,l} \text{trace}[Q] \right]$$

Final step, all these tensors should be symmetric and traceless. $\text{trace}[Q] = 0$, $Q_{k,l} = Q_{l,k}$

$$Q_{k,l} = c_2 \frac{4\pi}{15} * [2 Q_{k,l}]$$

$$c_2 = \frac{15}{8\pi}$$

What about rotational gradient?

The rotational gradient is described by

$$\mathcal{R} = \nabla_R = q \times \nabla_q$$

$$\mathcal{R}^2 = \nabla_R^2$$

And recall the property $\nabla_R^2 P_n = -n(n+1) P_n$

(Local) In[161]:=

```

RotOpSqrdd[ex_] :=
Module[{j1, j2, j3}, {j1 = getNewIndex[], j2 = getNewIndex[], j3 = getNewIndex[]};
(delta[j1, j2] - invr[2] * r[j1] * r[j2]) *
partialr[(delta[j3, j2] - invr[2] * r[j3] * r[j2]) * (partialr[ex, j3]) // Expand //
contract, j1] // contract];

```

(Local) In[162]:=

```

IrreducibleRanknTensor[1] // prettyprint
RotOpSqrdd[IrreducibleRanknTensor[1]] // prettyprint

IrreducibleRanknTensor[2] // prettyprint
RotOpSqrdd[IrreducibleRanknTensor[2]] // prettyprint;
% /. invr[x_] -> 1 // prettyprint
(*Note that  $r^2 = 1$  *)

IrreducibleRanknTensor[3] // prettyprint
RotOpSqrdd[IrreducibleRanknTensor[3]] // contract;
% /. invr[x_] -> 1 // prettyprint
(*Note that  $r^2 = 1$  *)

```

(Local) Out[162]=

 r_i

(Local) Out[163]=

$$-\frac{2 r_i}{r^2}$$

(Local) Out[164]=

$$r_i r_j - \frac{\delta_{i,j}}{3}$$

(Local) Out[166]=

$$-\frac{6 r_i r_j}{r^2} + 2 \delta_{i,j}$$

(Local) Out[167]=

$$r_i r_j r_k - \frac{1}{5} r_k \delta_{i,j} - \frac{1}{5} r_j \delta_{i,k} - \frac{1}{5} r_i \delta_{j,k}$$

(Local) Out[169]=

$$-12 r_i r_j r_k + \frac{12}{5} r_k \delta_{i,j} + \frac{12}{5} r_j \delta_{i,k} + \frac{12}{5} r_i \delta_{j,k}$$